Introduction to Minimal Surfaces

Carman Cater CCSU MATH 585 Spring 2022

History

- In 1762 Joseph-Louis Lagrange began investigating the problem of finding surfaces which minimize surface area given some boundary constraint in $\mathbb{R}^3$
  - This can be approached using calculus of variations

- Joseph Plateau (1801 - 1883) made the observation that soap films created by wire frames give surfaces of minimal area
  - Note the typical bubbles blown are not minimal surfaces.

In[1]:=

- Minimal surfaces are surfaces which locally minimize area
  - We will connect this with the equivalent condition that the mean curvature is zero everywhere, as well as link minimal surfaces to complex analysis and harmonic functions

In[2]:=

- In the early stages of development the only known minimal surfaces were the plane, helicoid, and catenoid
  - We now know many more: Scherk surface, Enneper surface, Costa’s surface, etc.

In[3]:=

- Here are a few soap film examples.

Out[4]=







If the two wire frames are pulled too far apart the soap film will split apart and form two disks in the wire frame. This is suggestive that there may not always be a minimal surface spanning two disconnected closed wire frames.

For simple closed curves there does exist a local area minimizing surface.

Surfaces with minimal area

We begin by deriving the minimal surface equation for surfaces in $\mathbb{R}^3$ given by $z = f(x, y)$.

Given some closed bounded domain D in $\mathbb{R}^2$ we take a surface

$G_f = \{ (x, y, f(x, y)) \mid (x, y) \in D \}$

and compute the coefficients of the first fundamental form.

$E = 1 + f_x^2$
$F = f_x f_y$
$G = 1 + f_y^2$

The area functional is given by

$\text{Area}(f) = \int_D \sqrt{EG - F^2}\, dA = \int_D \sqrt{1 + f_x^2 + f_y^2}\, dx\, dy = \int_D \sqrt{1 + |\nabla f|^2}\, dx\, dy$
Where $\nabla f = (f_x, f_y)$

Our goal is to find functions f that minimize this functional.

To do this we imagine deforming f by taking a function $h: D \to \mathbb{R}^3$ such that $h(\mathrm{bD}) = 0$ (vanishes on the boundary) and looking at $f + \mathrm{sh}$ where $s \in \mathbb{R}$.

From calculus of variations we have f is a stationary point of the area functional if and only if

$\frac{d}{ds} |_{s=0} \text{Area}(f + \mathrm{sh}) = 0$

$\int_D \frac{d}{ds} |_{s=0} \sqrt{1 + (f_x + \mathrm{sh}_x)^2 + (f_y + \mathrm{sh}_y)^2}\, dx\, dy = 0$

$\int_D \frac{f_x h_x + f_y h_y}{\sqrt{1 + |\nabla f|^2}}\, dx\, dy = 0$

Integrating both summands by parts (remember, h vanishes on bD) and factoring out an h allows us to conclude this integral is identically zero if and only if

$\left(1 + f_y^2\right) f_{xx} - 2 f_x f_y f_{xy} + \left(1 + f_x^2\right) f_{yy} = 0$

This PDE was discovered by Lagrange in ~1760 and is known as the minimal surface equation. At the time the only known solution was a plane.

Note in order to truly determine if this is a minimum and not a maximum, we would need to compute the second variation of the functional.

Geometric interpretation of the minimal surface equation

In this section we will connect the minimal surface equation

$$\left(1+f_y^2\right) f_{xx} - 2 f_x f_y f_{xy} + \left(1+f_x^2\right) f_{yy} = 0$$

to its mean curvature H.

We can use the formula for mean curvature given in our course

$$H = \frac{e\,G - 2 f\,F + g\,E}{E\,G - F^2}$$

where E, F, G, e, f, g are the coefficients of the first and second fundamental form.

Given some closed bounded domain D in $\mathbb{R}^2$ we take a surface

$$G_f = \{(x, y, f(x, y)) \mid (x, y) \in D\}$$

and compute the coefficients of the first and second fundamental form

$$E = 1 + f_x^2 \qquad e = \frac{f_{xx}}{\sqrt{1+|\nabla f|^2}}$$

$$F = f_x f_y \qquad f = \frac{f_{xy}}{\sqrt{1+|\nabla f|^2}}$$

$$G = 1 + f_y^2 \qquad g = \frac{f_{yy}}{\sqrt{1+|\nabla f|^2}}$$

Thus $H = \frac{eG - 2\,fF + gE}{EG - F^2} = \frac{\left(1+f_y^2\right) f_{xx} - 2 f_x f_y f_{xy} + \left(1+f_x^2\right) f_{yy}}{\left(EG - F^2\right)\sqrt{1+|\nabla f|^2}}$

Therefore a smooth surface f = f(x,y) in $\mathbb{R}^3$ satisfies the minimal surface equation if and only if its mean curvature equals zero (H = 0) at every point.

Conformal parametrization of a minimal surface

A mapping $X: D \to \mathbb{R}^n$ with D a subset of $\mathbb{R}^2$ is conformal if it preserves angles at every point. In our course this was studied in the section on isothermal coordinates.

We learned that for a conformal mapping F, we must have

$E = X_u \cdot X_u = X_v \cdot X_v = G$ and $F = X_u \cdot X_v = 0$

It turns out that locally near any given point in our domain D we can always replace a parametrization of an immersed surface by a conformal one.

This allows us to state our first theorem concerning minimal surfaces.

Theorem: A conformal immersion F(u, v) = (x, y, z): $D \to \mathbb{R}^3$ with D a subset of $\mathbb{R}^2$ is a minimal surface if and only if F is harmonic, i.e.

$\triangle F = (\triangle x, \triangle y, \triangle z) = (0, 0, 0)$

where for example, $\triangle x = x_{uu} + x_{vv}$ denotes the Laplace operator.

Let us verify by checking the Catenoid (formed by rotating a catenary around an axis).

```
In[5]:= F[u_, v_] = {Cosh[v] Cos[u], Cosh[v] Sin[u], v};
```

In[6]:=
```
ParametricPlot3D[F[u, v], {u, 0, 2 π}, {v, -2, 2}]
```

Out[6]=



Checking conformality of this parametrization and its Laplacian

In[7]:=
```
Simplify[D[F[u, v], u].D[F[u, v], u]] ==
 Simplify[D[F[u, v], v].D[F[u, v], v]]
```

Out[7]= True

In[8]:=
```
Simplify[D[F[u, v], u].D[F[u, v], v]]
```

Out[8]= 0

In[9]:=
```
Laplacian[{F[u, v][[1]], F[u, v][[3]], F[u, v][[3]]}, {u, v}]
```

Out[9]= {0, 0, 0}

Let us also check the Helicoid

In[10]:=
```
F[u_, v_] = {Sin[u] Sinh[v], -Cos[u] Sinh[v], u};
```

In[11]:=
```
ParametricPlot3D[F[u, v], {u, 0, 2 π}, {v, -2, 2}]
```

Out[11]=



Checking conformality of this parametrization and its Laplacian

In[12]:=
```
Simplify[D[F[u, v], u].D[F[u, v], u]] ==
 Simplify[D[F[u, v], v].D[F[u, v], v]]
```

Out[12]=
```
True
```

In[13]:=
```
Simplify[D[F[u, v], u].D[F[u, v], v]]
```

Out[13]=
```
0
```

In[14]:=
```
Laplacian[{F[u, v][[1]], F[u, v][[3]], F[u, v][[3]]}, {u, v}]
```

Out[14]=
```
{0, 0, 0}
```

Note the helicoid is the only ruled minimal surface.

Minimal surfaces in higher dimensional Euclidean spaces

In this section we see a few results regarding minimal surfaces in $\mathbb{R}^n$ with n ≥ 3.   F being a stationary point of the area functional, F being a harmonic immersion, as well as the mean curvature vector field **H** being zero everywhere.

Area functional: Area(F) = $\int_D \sqrt{\,|F_u|^2\,|F_v|^2 - |F_u \cdot F_v|^2}\,$ du dv

Theorem: Let F = $(F_1, F_2, \ldots, F_n) : D \to \mathbb{R}^n$ with D a subset of $\mathbb{R}^2$ be a conformal (angle preserving) immersion. F is a stationary point of the area functional if and only if F is harmonic.

Recall F being harmonic means $\triangle$F = $(\triangle F_1, \triangle F_2, \ldots, \triangle F_n)$ = (0, 0, ..., 0) where for example $\triangle F_1 = F_{1_{uu}} + F_{1_{vv}}$

Theorem: A conformal immersion F: $D \to \mathbb{R}^n$ satisfies

$$\triangle F = \frac{1}{2} \mid \nabla F \mid^2 \boldsymbol{H}$$

where $\mid \nabla F \mid^2 = \mid F_u \mid^2 + \mid F_v \mid^2$ and **H** is the mean curvature vector given by **H** = $\sum^n{}_{i=3} H_i \, \mathbf{e}_i$

In particular, the mean curvature vector field **H** is zero if and only if F is harmonic, if and only if F is a stationary point of the area functional.

Complex analytic view of minimal surfaces

Take a function f(z) = u(x, y) + i v(x, y) from $\mathbb{C} \to \mathbb{C}$.

where if we write

z = x + i y,   zbar = x - i y

we can solve for x and y

$x = \frac{1}{2}(z + \text{zbar})$ and $y = \frac{1}{2i}(z - \text{zbar})$

Recall we say f is holomorphic to mean it is complex differentiable. This turns out (if we assume continuous second partial derivatives) to be equivalent to satisfying the Cauchy-Riemann equations

$u_x = v_y$ and $u_y = -v_x$

Definition: A holomorphic immersion $F = (F_1, F_2, \ldots, F_n) : D \to \mathbb{C}^n$ where $D \subset \mathbb{C}$ for $n \geq 3$ is called a holomorphic null curve if it satisfies the nullity condition $(F_1')^2 + (F_2')^2 + \ldots + (F_n')^2 = 0$ where the derivatives are with respect to z.

It turns out that if $Z = X + i\,Y : D \to \mathbb{C}^n$ is a holomorphic null curve then its real part $X = \Re Z$ and its imaginary part $Y = \Im Z$ are conformal minimal surfaces.

$X^t(z) = \Re\left(e^{it} Z(z)\right) : D \to \mathbb{R}^n$ are called associated minimal surfaces.

Let us look at an example of the Helicatenoid.

```
In[15]:= Z[z_] = {Cos[z], Sin[z], -I z};
```

Note this is holomorphic. We now check the nullity condition

```
In[16]:= D[Z[z], z]
Out[16]= {-Sin[z], Cos[z], -i}
In[17]:= Simplify[D[Z[z], z].D[Z[z], z]]
Out[17]= 0
```

Getting in terms of u-v coordinates we have

```
In[18]:= ComplexExpand[Re[(Cos[t] + I Sin[t]) Z[u + I v]]]
Out[18]= {Cos[t] Cos[u] Cosh[v] + Sin[t] Sin[u] Sinh[v],
          Cos[t] Cosh[v] Sin[u] - Cos[u] Sin[t] Sinh[v], v Cos[t] + u Sin[t]}
In[19]:= X[t_, u_, v_] = Cos[t] {Cos[u] Cosh[v], Cosh[v] Sin[u], v} +
            Sin[t] {Sin[u] Sinh[v], -Cos[u] Sinh[v], u};
In[20]:= Manipulate[ParametricPlot3D[X[t, u, v], {u, 0, 2 π}, {v, -2, 2},
           PlotPoints → 10, MaxRecursion → 2, Mesh → 3], {t, 0, π}]
```





Examples of minimal surfaces using Enneper-Weierstrass Parametrization

Take a holomorphic function f and meromorphic function g (meaning g is holomorphic except for a set of isolated points).

A result due to Enneper and Weierstrass states that

$$\mathcal{R}\int \begin{pmatrix} f\left(1-g^2\right) \\ i\,f\left(1+g^2\right) \\ 2\,f\,g \end{pmatrix} dz$$

gives a minimal surface. Let us use this to explore a few well known examples.

Bour's Surface

In[21]:=
```
x[f_, g_] = {f (1 - g^2), I f (1 + g^2), 2 f g};
```

In[22]:=
```
p = x[1, √z];
```

In[23]:=
```
y[z_] = {Integrate[p[[1]], z],
   Integrate[p[[2]], z], Integrate[p[[3]], z]}
```

Out[23]= $\left\{z-\frac{z^2}{2},\ i\left(z+\frac{z^2}{2}\right),\ \frac{4\,z^{3/2}}{3}\right\}$

In[24]:=
```
m[u_, v_] = ComplexExpand[Re[y[u + I v]]]
```

Out[24]= $\left\{u-\frac{u^2}{2}+\frac{v^2}{2},\ -v-u\,v,\ \frac{4}{3}\left(u^2+v^2\right)^{3/4}\mathrm{Cos}\left[\frac{3}{2}\mathrm{Arg}[u+i\,v]\right]\right\}$

In[25]:=
```
n = Simplify[m[r Cos[θ], r Sin[θ]]]
```

Out[25]= $\left\{r\,\mathrm{Cos}[\theta]-\frac{1}{2}r^2\,\mathrm{Cos}[2\,\theta],\ -r\,(1+r\,\mathrm{Cos}[\theta])\,\mathrm{Sin}[\theta],\ \frac{4}{3}\left(r^2\right)^{3/4}\mathrm{Cos}\left[\frac{3}{2}\mathrm{Arg}[r\,(\mathrm{Cos}[\theta]+i\,\mathrm{Sin}[\theta])]\right]\right\}$

In[26]:=
```
ParametricPlot3D[n, {θ, 0, 2 π}, {r, 0, 10}, PlotPoints → 50]
```

Out[26]=



Enneper's Surface

In[27]:=
```
x[f_, g_] = {f (1 - g^2), I f (1 + g^2), 2 f g};
```

In[28]:=
```
p = x[1, z];
```

In[29]:=
```
y[z_] = {Integrate[p[[1]], z],
  Integrate[p[[2]], z], Integrate[p[[3]], z]}
```

Out[29]= $\left\{z - \frac{z^3}{3},\ i \left(z + \frac{z^3}{3}\right),\ z^2\right\}$

In[30]:=
```
m[u_, v_] = ComplexExpand[Re[y[u + I v]]]
```

Out[30]= $\left\{u - \frac{u^3}{3} + u\, v^2,\ -v - u^2\, v + \frac{v^3}{3},\ u^2 - v^2\right\}$

In[31]:=
```
n = Simplify[m[r Cos[θ], r Sin[θ]]]
```

Out[31]= $\left\{-\frac{1}{3}\, r\, \mathrm{Cos}[\theta]\, \left(-3 - r^2 + 2\, r^2\, \mathrm{Cos}[2\,\theta]\right),\right.$
$\left. -\frac{1}{3}\, r\, \left(3 + r^2 + 2\, r^2\, \mathrm{Cos}[2\,\theta]\right)\, \mathrm{Sin}[\theta],\ r^2\, \mathrm{Cos}[2\,\theta]\right\}$

In[32]:=
```
ParametricPlot3D[n, {θ, -π, π},
  {r, 0, 1.5}, PlotPoints → 50, MaxRecursion → 4]
```

Out[32]=



Henneberg's Surface

In[33]:=
```
x[f_, g_] = {f (1 - g^2), I f (1 + g^2), 2 f g};
```

In[34]:=
```
p = x[2 - 2 z^-4, z];
```

In[35]:=
```
y[z_] = {Integrate[p[[1]], z],
   Integrate[p[[2]], z], Integrate[p[[3]], z]}
```

Out[35]=

$$\left\{\frac{2}{3z^3} - \frac{2}{z} + 2z - \frac{2z^3}{3},\ i\left(\frac{2}{3z^3} + \frac{2}{z} + 2z + \frac{2z^3}{3}\right),\ \frac{2}{z^2} + 2z^2\right\}$$

In[36]:= `m[u_, v_] = ComplexExpand[Re[y[u + I v]]]`

Out[36]=
$$\left\{2u - \frac{2u^3}{3} + 2uv^2 + \frac{2u^3}{3(u^2+v^2)^3} - \frac{2uv^2}{(u^2+v^2)^3} - \frac{2u}{u^2+v^2},\right.$$
$$-2v - 2u^2 v + \frac{2v^3}{3} + \frac{2u^2 v}{(u^2+v^2)^3} - \frac{2v^3}{3(u^2+v^2)^3} + \frac{2v}{u^2+v^2},$$
$$\left.2u^2 - 2v^2 + \frac{2u^2}{(u^2+v^2)^2} - \frac{2v^2}{(u^2+v^2)^2}\right\}$$

In[37]:= `n = Simplify[m[r Cos[θ], r Sin[θ]]]`

Out[37]=
$$\left\{\frac{1}{3r^3} 2\,\mathrm{Cos}[\theta]\left(3r^2(-1+r^2) - (-1+r^6)\,\mathrm{Cos}[\theta]^2 + 3(-1+r^6)\,\mathrm{Sin}[\theta]^2\right),\right.$$
$$-\frac{1}{3r^3} 2(-1+r^2)\left(1+4r^2+r^4+2(1+r^2+r^4)\,\mathrm{Cos}[2\theta]\right)\mathrm{Sin}[\theta],$$
$$\left.\frac{2(1+r^4)\,\mathrm{Cos}[2\theta]}{r^2}\right\}$$

In[38]:= `ParametricPlot3D[n, {θ, 0, 2π}, {r, 0, 1}, PlotPoints → 150, RegionFunction → (#1^2 + #2^2 + #3^2 < 10^2 &)]`

Out[38]=


Scherks second surface

In[39]:= $x[f\_, g\_] = \{f\,(1 - g^2),\ I\,f\,(1 + g^2),\ 2\,f\,g\};$

In[40]:= $p = x\left[\frac{4}{1 - z^4},\ I\,z\right];$

In[41]:= y[z_] = {Integrate[p[[1]], z], Integrate[p[[2]], z], Integrate[p[[3]], z]}

Out[41]= $\left\{-4\left(\frac{1}{2}\,\mathrm{Log}[1 - z] - \frac{1}{2}\,\mathrm{Log}[1 + z]\right),\ 4\,i\,\mathrm{ArcTan}[z],\ -8\,i\left(\frac{1}{4}\,\mathrm{Log}\left[1 - z^2\right] - \frac{1}{4}\,\mathrm{Log}\left[1 + z^2\right]\right)\right\}$

In[42]:= m[u_, v_] = ComplexExpand[Re[y[u + I v]]]

Out[42]= $\left\{-\mathrm{Log}\left[(1 - u)^2 + v^2\right] + \mathrm{Log}\left[(1 + u)^2 + v^2\right],\ \mathrm{Log}\left[u^2 + (1 - v)^2\right] - \mathrm{Log}\left[u^2 + (1 + v)^2\right],\ 2\,\mathrm{Arg}\left[1 - (u + i\,v)^2\right] - 2\,\mathrm{Arg}\left[1 + (u + i\,v)^2\right]\right\}$

In[43]:= n = Simplify[m[r Cos[θ], r Sin[θ]]]

Out[43]= $\left\{-\mathrm{Log}\left[1 + r^2 - 2\,r\,\mathrm{Cos}[\theta]\right] + \mathrm{Log}\left[1 + r^2 + 2\,r\,\mathrm{Cos}[\theta]\right],\ \mathrm{Log}\left[1 + r^2 - 2\,r\,\mathrm{Sin}[\theta]\right] - \mathrm{Log}\left[1 + r^2 + 2\,r\,\mathrm{Sin}[\theta]\right],\ 2\left(\mathrm{Arg}\left[1 - r^2\,(\mathrm{Cos}[\theta] + i\,\mathrm{Sin}[\theta])^2\right] - \mathrm{Arg}\left[1 + (r\,\mathrm{Cos}[\theta] + i\,r\,\mathrm{Sin}[\theta])^2\right]\right)\right\}$

In[44]:= `ParametricPlot3D[n, {θ, 0, 2 π}, {r, 0, 1}, PlotPoints → 150]`

Out[44]=



More Examples!

All of the examples we've seen trace back to the 18th and 19th centuries.

This is still an active area of research. There is a 1 hour long movie from ~1980's (found on YouTube) called Natural Minimal Surfaces Via Theory and Computation by David Hoffman of (previously) UMass Amherst.

Here are some pictures of more modern 20th century minimal surfaces.

Gyroid


Saddle Tower



Costa-Hoffman-Meeks




The first nine Chen-Gackstatter surfaces



Works Cited

- Minimal surfaces for undergraduates by Franc Forstneric
- Wikipedia (read many pages)
- Wolfram MathWorld (read many pages)
- A mathematical development of minimal surface theory: From soap films to black holes by Timothy Pitts
- What are Minimal surfaces? by Rukmini Dey
- The Weierstrass representation always gives a minimal surface by Roshan Sharma
- https://complex-analysis.com
- Natural Minimal Surfaces Via Theory and Computation, David Hoffman
- Images:
  - https://plus.maths.org/content/sites/plus.maths.org/files/helicoid_carousel.jpg
  - https://wdjoyner.files.wordpress.com/2018/04/douglas_soap-film1.jpg?w=584&h=390
  - https://www.soapbubble.dk/content/2-artikler/5-former/dsc_1503.jpg
  - https://mathworld.wolfram.com/images/gifs/GyroidSculpture.jpg